vv_v#:"*********"*25<           01 = x coord
8,:  >              ^           02 = y coord
0>^  ^"*     * *"*25<
1    >              ^
p    ^"* *** * *"*25<
2    >              ^
0    ^"* *     *"*25<
2    >              ^
p    ^"* * *   *"*25<
"    >              ^       
O    ^"* ***** *"*25<       >,v
"    >              ^       |:<"You hit a wall! Game over!"0<
0    ^"*     * *"*25<       >25*,@                          |-*84gg20g10<
1    >              ^v ,*62                       pg20g10"O"<  <       
g    ^"*   *   *"*25<                                  >00g"w"-|
0    >              ^                          >00g"e"-|       >01g1-01p^
2    ^"*********"*250<                 >00g"s"-|       >01g1+01p        ^
g  > " "01g02gp         "?",~~$:00p"n"-|       >02g2+02p                ^
>p              62*, ^                 >02g2-02p                        ^
